Isabelle/HOL Exercises
Logic and Sets

Predicate Logic

We are again talking about proofs in the calculus of Natural Deduction. In addition to the rules given in the exercise “Propositional Logic”, you may now also use

exI: $P\ x \Longrightarrow \exists x.\ P\ x$
exE: $[\![\exists x.\ P\ x;\ \bigwedge x.\ P\ x \Longrightarrow Q]\!] \Longrightarrow Q$
allI: $(\bigwedge x.\ P\ x) \Longrightarrow \forall x.\ P\ x$
allE: $[\![\forall x.\ P\ x;\ P\ x \Longrightarrow R]\!] \Longrightarrow R$

Give a proof of the following propositions or an argument why the formula is not valid:

```
lemma "(∃x. ∀y. P x y) ⟶ (∀y. ∃x. P x y)"
  apply (rule impI)
  apply (erule exE)
  apply (rule allI)
  apply (erule allE)
  apply (rule exI)
  apply assumption
done
```

```
lemma "(∀x. P x ⟶ Q) = ((∃x. P x) ⟶ Q)"
  apply (rule iffI)

  apply (rule impI)
  apply (erule exE)
  apply (erule allE)
  apply (erule impE)
  apply assumption+

  apply (rule allI)
  apply (rule impI)
  apply (erule impE)
  apply (rule exI)
  apply assumption+
done
```

```
lemma "((∀ x. P x) ∧ (∀ x. Q x)) = (∀ x. (P x ∧ Q x))"
  apply (rule iffI)

  apply (erule conjE)
  apply (rule allI)
  apply (erule allE)+
  apply (rule conjI)
  apply assumption+

  apply (rule conjI)
  apply (rule allI)
  apply (erule allE)
  apply (erule conjE)
  apply assumption
  apply (rule allI)
  apply (erule allE)
  apply (erule conjE)
  apply assumption
done

lemma "((∀ x. P x) ∨ (∀ x. Q x)) = (∀ x. (P x ∨ Q x))"
  refute
⋮
```

A possible counterexample is: `P = even`, `Q = odd`, interpreted over the natural numbers.

```
lemma "((∃ x. P x) ∨ (∃ x. Q x)) = (∃ x. (P x ∨ Q x))"
  apply (rule iffI)

  apply (erule disjE)
  apply (erule exE)
  apply (rule exI)
  apply (rule disjI1)
  apply assumption

  apply (erule exE)
  apply (rule exI)
  apply (rule disjI2)
  apply assumption

  apply (erule exE)
  apply (erule disjE)
  apply (rule disjI1)
  apply (rule exI)
```

```
  apply assumption

  apply (rule disjI2)
  apply (rule exI)
  apply assumption
done

lemma "(∀x. ∃y. P x y) ⟶ (∃y. ∀x. P x y)"
  refute
⋮
```

For a possible counterexample, let `P x y` be the statement "`y` is successor of `x`", interpreted over the natural numbers.

```
lemma "(¬ (∀ x. P x)) = (∃ x. ¬ P x)"
  apply (rule iffI)

  apply (rule classical)
  apply (erule notE)
  apply (rule allI)
  apply (rule classical)
  apply (erule notE)
  apply (rule exI)
  apply assumption

  apply (erule exE)
  apply (rule notI)
  apply (erule allE)
  apply (erule notE)
  apply assumption
done
```